
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.3.2.1 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-06_D19wk38’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
8.1.3.2.1
Test Group:
\NR5GC\8_1_3\
ATS Version:
iwd-TTCN3-B2019-06_D19wk38
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for NR5GC test case 8.1.3.2.1 on wk38 ATS

4.1 Change 1

	Function name
	f_MTC_ConnectPTCs_NR5GC_IRAT()

	Reason for change
	IMS port on NR PTC is not currently connected for IRAT test cases, causing an error inside function f_NR_DelayForUserPlaneSignalling()

	Summary of change
	Connect IPCAN port on IMS PTC to IMS port on NR PTC, and OtherIPCAN port on IMS PTC to IMS port on EUTRA PTC (following model used previously for EUTRA IRAT test cases)

	TTCN module
	MTC_Main_NR5GC_IRAT

	MCC160 Comment
	


Before change

function f_MTC_ConnectPTCs_NR5GC_IRAT(SYSTEM_NR5GC_IRAT p_System,

                                  NR5GC_PTC p_Nr,

                                  EUTRA_5GS_PTC p_Eutra,

                                  IMS_PTC p_ImsPdn1,

                                  IMS_PTC p_ImsPdn2) runs on MTC_NR5GC_IRAT

  {

    var NASEMU_EUTRA_PTC   v_NASEMU_EUTRA_PTC;

    var NASEMU_NRNG_PTC v_NASEMU_NR_PTC;

    var IP_PTC v_IP_PTC;

    var IMS_PTC v_ImsPdn1;

    vc_Components.Nr    := p_Nr;

    vc_Components.Eutra := p_Eutra;

    if ((p_Nr == null) or (p_Eutra == null)) {

      FatalError (__FILE__, __LINE__, "invalid configuration");

      mtc.stop;

    }

    // ***** MTC *****

    f_MTC_BASE_Map(p_System);

    // ***** NR PTC *****

    f_NR_BASE_PTC_Map(p_System, p_Nr);

    v_NASEMU_NR_PTC := f_NasEmu5G_CreateConnectAndMap_NRNG(p_System, p_Nr);  // create NasEmu, map system ports and connect with NR PTC

    v_NASEMU_NR_PTC.start(f_NASEMU_MainLoop_NRNG());                            // start NASEMU

    // ***** EUTRA PTC *****

    f_EUTRA_PTC_Map(p_System, p_Eutra);                // map EUTRA system ports

    v_NASEMU_EUTRA_PTC := f_NasEmu_CreateConnectAndMap_EUTRA(p_System, p_Eutra);  // create NasEmu, map system ports and connect with EUTRA PTC

    v_NASEMU_EUTRA_PTC.start(f_NASEMU_MainLoop_EUTRA());                          // start NASEMU

    // ***** Create, map, connect and start IP PTC *****

    v_IP_PTC := f_IP_PTC_CreateAndMap(p_System);

    connect(p_Eutra:IP, v_IP_PTC:EUTRA_CTRL);

    connect(p_Nr:IP, v_IP_PTC:NR_CTRL);

    v_IP_PTC.start(f_IP_PTC_MainLoop());

    // ***** Create, connect and start IMS PTC *****

    // IMS for PDN1: if PTC is not created yet the default IMS PTC will be used

    if (p_ImsPdn1 == null) { v_ImsPdn1 := f_IMS_PTC_Create(); }         // => default IMS handling for PDN1

    else                   { v_ImsPdn1 := p_ImsPdn1; }

    connect(v_ImsPdn1:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN1]);

    connect(v_ImsPdn1:MMI, mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1]);

    connect(v_ImsPdn1:IPCAN, p_Eutra:IMS[tsc_Index_PDN1]);
    if (p_ImsPdn1 == null) {         /* => we have no test case specific IMS implementation */

      v_ImsPdn1.start(f_IMS_DefaultHandler());

    }

    // IMS for PDN (emergency call)

    if (p_ImsPdn2 != null) {                             // test case specific IMS handling for PDN2

      connect(p_ImsPdn2:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IPCAN,         p_Eutra:IMS[tsc_Index_PDN2]);

      connect(p_ImsPdn2:MMI,           mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2]);

    }

    // ***** Connect PTCs to MTC *****

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_NR],       p_Nr:UT);

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_EUTRA],    p_Eutra:UT);

    // ***** Connect NR PTC to other PTCs *****

    connect(p_Nr:EUTRA,    p_Eutra:NR);

    f_UT_InitialPowerOffUE(Ut);

  } 
After change

   function f_MTC_ConnectPTCs_NR5GC_IRAT(SYSTEM_NR5GC_IRAT p_System,

                                  NR5GC_PTC p_Nr,

                                  EUTRA_5GS_PTC p_Eutra,

                                  IMS_PTC p_ImsPdn1,

                                  IMS_PTC p_ImsPdn2) runs on MTC_NR5GC_IRAT

  {

    var NASEMU_EUTRA_PTC   v_NASEMU_EUTRA_PTC;

    var NASEMU_NRNG_PTC v_NASEMU_NR_PTC;

    var IP_PTC v_IP_PTC;

    var IMS_PTC v_ImsPdn1;

    vc_Components.Nr    := p_Nr;

    vc_Components.Eutra := p_Eutra;

    if ((p_Nr == null) or (p_Eutra == null)) {

      FatalError (__FILE__, __LINE__, "invalid configuration");

      mtc.stop;

    }

    // ***** MTC *****

    f_MTC_BASE_Map(p_System);

    // ***** NR PTC *****

    f_NR_BASE_PTC_Map(p_System, p_Nr);

    v_NASEMU_NR_PTC := f_NasEmu5G_CreateConnectAndMap_NRNG(p_System, p_Nr);  // create NasEmu, map system ports and connect with NR PTC

    v_NASEMU_NR_PTC.start(f_NASEMU_MainLoop_NRNG());                            // start NASEMU

    // ***** EUTRA PTC *****

    f_EUTRA_PTC_Map(p_System, p_Eutra);                // map EUTRA system ports

    v_NASEMU_EUTRA_PTC := f_NasEmu_CreateConnectAndMap_EUTRA(p_System, p_Eutra);  // create NasEmu, map system ports and connect with EUTRA PTC

    v_NASEMU_EUTRA_PTC.start(f_NASEMU_MainLoop_EUTRA());                          // start NASEMU

    // ***** Create, map, connect and start IP PTC *****

    v_IP_PTC := f_IP_PTC_CreateAndMap(p_System);

    connect(p_Eutra:IP, v_IP_PTC:EUTRA_CTRL);

    connect(p_Nr:IP, v_IP_PTC:NR_CTRL);

    v_IP_PTC.start(f_IP_PTC_MainLoop());

    // ***** Create, connect and start IMS PTC *****

    // IMS for PDN1: if PTC is not created yet the default IMS PTC will be used

    if (p_ImsPdn1 == null) { v_ImsPdn1 := f_IMS_PTC_Create(); }         // => default IMS handling for PDN1

    else                   { v_ImsPdn1 := p_ImsPdn1; }

    connect(v_ImsPdn1:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN1]);

    connect(v_ImsPdn1:MMI, mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1]);

    connect(v_ImsPdn1:OtherIPCAN, p_Eutra:IMS[tsc_Index_PDN1]); 
    connect(v_ImsPdn1:IPCAN, p_Nr:IMS[tsc_Index_PDN1]);
    if (p_ImsPdn1 == null) {         /* => we have no test case specific IMS implementation */

      v_ImsPdn1.start(f_IMS_DefaultHandler());

    }

    // IMS for PDN (emergency call)

    if (p_ImsPdn2 != null) {                             // test case specific IMS handling for PDN2

      connect(p_ImsPdn2:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN2]);

      connect(p_ImsPdn2:OtherIPCAN,         p_Eutra:IMS[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IPCAN,         p_Nr:IMS[tsc_Index_PDN2]); 

      connect(p_ImsPdn2:MMI,           mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2]);

    }

    // ***** Connect PTCs to MTC *****

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_NR],       p_Nr:UT);

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_EUTRA],    p_Eutra:UT);

    // ***** Connect NR PTC to other PTCs *****

    connect(p_Nr:EUTRA,    p_Eutra:NR);

    f_UT_InitialPowerOffUE(Ut);

  }
4.2 Change 2

	Function name
	f_NR5GC_CellInfo_SetSIB5_FreqListAllCells()

	Reason for change
	Stored cell information for some cells will not be available if the corresponding frequencies are not defined, giving a fatal error in the execution of f_NR5GC_CellInfo_SetSIB5()

	Summary of change
	Check the corresponding frequency is defined before initialising SIB5 info

	TTCN module
	NR5GC_IRAT_CommonFunctions

	MCC160 Comment
	


Before change

    function f_NR5GC_CellInfo_SetSIB5_FreqListAllCells(template (value) SIB5 p_SIB5) runs on NR_BASE_PTC

  {

    f_NR5GC_CellInfo_SetSIB5(nr_Cell1, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell2, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell4, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell11, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell3, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell12, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell23, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell6, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell13, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell14, p_SIB5);
  }
After change

   function f_NR5GC_CellInfo_SetSIB5_FreqListAllCells(template (value) SIB5 p_SIB5) runs on NR_BASE_PTC

  {

    //KS R5sxxxxx

    var template(value) NR_Frequencyf1tof4_Type v_NR_PrimaryFrequencyf1Tof4 := f_NR_GetFrequencySA_f1tof4(px_NR_PrimaryBand);
    f_NR5GC_CellInfo_SetSIB5(nr_Cell1, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell2, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell4, p_SIB5);

    f_NR5GC_CellInfo_SetSIB5(nr_Cell11, p_SIB5);   

    if (ispresent(v_NR_PrimaryFrequencyf1Tof4.FrequencyInfoDL_f2)){

      f_NR5GC_CellInfo_SetSIB5(nr_Cell3, p_SIB5);

      f_NR5GC_CellInfo_SetSIB5(nr_Cell12, p_SIB5);

      f_NR5GC_CellInfo_SetSIB5(nr_Cell23, p_SIB5);

    }

    if (ispresent(v_NR_PrimaryFrequencyf1Tof4.FrequencyInfoDL_f3)){

      f_NR5GC_CellInfo_SetSIB5(nr_Cell6, p_SIB5);

      f_NR5GC_CellInfo_SetSIB5(nr_Cell13, p_SIB5);

    }

    if (ispresent(v_NR_PrimaryFrequencyf1Tof4.FrequencyInfoDL_f4)){

      f_NR5GC_CellInfo_SetSIB5(nr_Cell14, p_SIB5);

    }

  } 
4.3 Change 3

	Function name
	fl_NR5GC_IRAT_InitSysinfo()

	Reason for change
	The function f_EUTRA_NR_SysinfoCoOrd() calls f_EUTRA_NR_WaitForCoordSysInfo() which expects NR system information, which will give a mismatch if called from NR PTC.


	Summary of change
	Replace call to f_EUTRA_NR_SysInfoCoOrd() with new function f_NR_EUTRA_SysInfoCoOrd() (see change 4)

	TTCN module
	NR5GC_IRAT_CommonFunctions

	MCC160 Comment
	


Before change

  function fl_NR5GC_IRAT_InitSysinfo(NR_SysinfoCombination_Type p_NR_SysinfoCombination) runs on NR_BASE_PTC
  {

    var template (value) EUTRA_NR_CoOrd_SysInfo_Type v_EUTRA_SysInfoList;

    var template (value) EUTRA_NR_CoOrd_SysInfo_Type v_NRSysInfoList := f_NR5GC_IRAT_ListOfFrequencies(); //Get NR frequencies to send to EUTRA

    var template (value) SIB5 v_SIB5;
    // Get EUTRA frequency info and send NR frequencies

    v_EUTRA_SysInfoList := f_EUTRA_NR_SysinfoCoOrd(EUTRA, v_NRSysInfoList);
    select (p_NR_SysinfoCombination){

        case (NR_6, NR_7){

            v_SIB5 := f_NR5GC_IRAT_InitSIB5(v_EUTRA_SysInfoList);

            f_NR5GC_CellInfo_SetSIB5_FreqListAllCells(v_SIB5);

        }

        case else {} //do nothing
     }

   }

After change

   function fl_NR5GC_IRAT_InitSysinfo(NR_SysinfoCombination_Type p_NR_SysinfoCombination) runs on NR_BASE_PTC

  {

    var template (value) EUTRA_NR_CoOrd_SysInfo_Type v_EUTRA_SysInfoList;

    var template (value) EUTRA_NR_CoOrd_SysInfo_Type v_NRSysInfoList := f_NR5GC_IRAT_ListOfFrequencies(); //Get NR frequencies to send to EUTRA

    var template (value) SIB5 v_SIB5;

    // Get EUTRA frequency info and send NR frequencies

   //KS R5sxxx

      v_EUTRA_SysInfoList := f_NR_EUTRA_SysinfoCoOrd(EUTRA, v_NRSysInfoList);
    select (p_NR_SysinfoCombination){

        case (NR_6, NR_7){

            v_SIB5 := f_NR5GC_IRAT_InitSIB5(v_EUTRA_SysInfoList);

            f_NR5GC_CellInfo_SetSIB5_FreqListAllCells(v_SIB5);

        }

        case else {} //do nothing

    }

   }
4.4 Change 4

	Function name
	new function f_NR_EUTRA_SysinfoCoOrd

	Reason for change
	The function f_EUTRA_NR_SysinfoCoOrd() calls f_EUTRA_NR_WaitForCoordSysInfo() which expects NR system information, which will give a mismatch if called from NR PTC.


	Summary of change
	New frunction for coordinating sys info from NR PTC

	TTCN module
	EUTRA_NR_CoordMsg

	MCC160 Comment
	


New function

  function f_NR_EUTRA_SysinfoCoOrd(EUTRA_NR_CO_ORD_PORT p_Port,

                                   template(value) EUTRA_NR_CoOrd_SysInfo_Type p_EUTRA_NR_CoOrd_SysInfo

                                   ) return EUTRA_NR_CoOrd_SysInfo_Type

  {

    var EUTRA_NR_CoOrd_SysInfo_Type v_SysInfoRx;

    p_Port.send(cms_EUTRA_NR_SysInfoEUTRA(p_EUTRA_NR_CoOrd_SysInfo));

    v_SysInfoRx := f_NR_WaitForCoOrd_SysInfoEUTRA (p_Port);

    return v_SysInfoRx;

  }

4.5 Change 5

	Function name
	f_NR_WaitForCoOrd_SysInfoEUTRA()

	Reason for change
	The function f_EUTRA_NR_SysinfoCoOrd() calls f_EUTRA_NR_WaitForCoordSysInfo() which expects NR system information, which will give a mismatch if called from NR PTC.


	Summary of change
	New functions to receive EUTRA sys info at NR PTC

	TTCN module
	EUTRA_NR_CoordMsg

	MCC160 Comment
	


New function

  function f_NR_WaitForCoOrd_SysInfoEUTRA(EUTRA_NR_CO_ORD_PORT p_Port) return EUTRA_NR_CoOrd_SysInfo_Type

  {

    var EUTRA_NR_Coordination_MSG v_ReceivedMsg;

    alt {

      [] p_Port.receive(cmr_EUTRA_NR_SysInfoEUTRA) -> value v_ReceivedMsg {}

      [] a_EUTRA_NR_Default(p_Port) {}

    }

    return v_ReceivedMsg.SysInfo;

  }
4.6 Change 6
	Function name
	f_TC_8_1_3_2_1_NR5GC()

	Reason for change
	1. Wrong INIT function is used
2. Cell release on EUTRA PTC is not triggered

	Summary of change
	1.use f_NR5GC_IRAT_Init(NR_6) for IRAT case 

2. send CoOrd trigger to EUTRA PTC after postamble

	TTCN module
	RRC_Measurement_NR5GC_NR

	MCC160 Comment
	


Before change

  function f_TC_8_1_3_2_1_NR5GC() runs on NR5GC_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B1 / Measurement of E-UTRA cells

    f_NR5GC_Init(NR_6);
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -85, -85); //Maximum power level for FR2 FFS

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(true);

    f_TC_8_1_3_2_1_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  }// End of f_TC_8_1_3_2_1_NR5GC()

After change

   function f_TC_8_1_3_2_1_NR5GC() runs on NR5GC_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B1 / Measurement of E-UTRA cells

    f_NR5GC_IRAT_Init(NR_6);

    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -85, -85); //Maximum power level for FR2 FFS

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(true);

    f_TC_8_1_3_2_1_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger("Postamble"));

  }// End of f_TC_8_1_3_2_1_NR5GC()
4.7 Change 7
	Function name
	f_TC_8_1_3_2_1_TestBody()

	Reason for change
	1.Currently used function f_EUTRA_NR_WaitForCoOrd_SysInfo() expects NR system information
2.Incorrect value used for RSRP threshold
3. Configured MeasGapConfig should be as 38.508-1 table 4.6.3-70 without conditiond GAP_FR1 or GAP_FR2 defined.

	Summary of change
	1, call new function f_NR_WaitForCoOrd_SysInfoEUTRA() (see change 5)

2.v_RsrpThres should be 60 corresponding to -80dbm
Note : An associated prose CR on 36.523-1 will be raised at RAN5#86

3. Use new template cs_MeasGapConfigUE instead of cs_MeasGapConfig_FR1

	TTCN module
	RRC_Measurement_NR5GC_NR

	MCC160 Comment
	


Before change

  function f_TC_8_1_3_2_1_TestBody() runs on NR_BASE_PTC

  {

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0_T1;

    var NR_AbsoluteCellPower_Type v_CellPower_T1_FR1 := -85;

    var PhysCellId v_PhysCellId;

    var ARFCN_ValueEUTRA v_ARFCN_ValueEUTRA;

    var EUTRA_PhysCellId v_EUTRA_CellId;

    var integer v_MeasurementBandwidth_int;

    var EUTRA_AllowedMeasBandwidth v_MeasurementBandwidth;

    var RSRP_Range v_RsrpThres;

    var Hysteresis v_Hysteresis := 4;

    var EventTriggerConfigInterRAT.reportAmount v_ReportAmount := r1;

    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList_NR;

    var template (value) MeasObjectParamsList_Type_EUTRA v_MeasuredObjectList_EUTRA;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    //Power levels of NR cell 1: Table 8.1.3.1.1.3.2-1, Table 8.1.3.1.1.3.2-2

    /*         |    FR1     |    FR2

     * -------------------------------

     *   T0    |    -85     |    FFS

     * -------------------------------

     *   T1    |    -85     |    FFS

     *

     */

    v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo (EUTRA);

    v_ARFCN_ValueEUTRA := v_CoOrd_SysInfo.Eutra[0].Arfcn;

    v_EUTRA_CellId := bit2int (v_CoOrd_SysInfo.Eutra[0].CellId);

    v_MeasurementBandwidth_int := v_CoOrd_SysInfo.Eutra[0].DL_Bandwidth;

    v_MeasurementBandwidth := f_NR5GC_IRAT_ConvertIntegerToEUTRA_AllowedMeasBandwidth(v_MeasurementBandwidth_int);

    v_CellPowerList_AtT0_T1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_T1_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    f_NR_SetCellPowerList(v_CellPowerList_AtT0_T1);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-RAT measurement and reporting for event B1

    //The UE transmits an RRCReconfigrationComplete message

    v_RsrpThres := 76; //@sic R5-195345 sic@

    v_MeasuredObjectList_NR := {cs_MeasObjectId1(nr_Cell1)};

    v_MeasuredObjectList_EUTRA := {cs_MeasObjectId2_EUTRA(v_EUTRA_CellId, v_ARFCN_ValueEUTRA)};

    v_ReportConfigList := {cs_NR_ReportConfig_reportConfigInterRAT(tsc_NR_IdReportConfigId1, cs_NR_ReportConfigInterRAT_Trigger_eventB1_EUTRA(v_RsrpThres, v_Hysteresis, v_ReportAmount))};

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR_EUTRA(v_ARFCN_ValueEUTRA, v_MeasurementBandwidth, -, v_MeasuredObjectList_NR, v_MeasuredObjectList_EUTRA, v_ReportConfigList, v_MeasIdConfigList, cs_MeasGapConfig_FR1, cs_38508_QuantityConfig_EUTRA);

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig);

    //@siclog "Steps  3" siclog@

    //Check: Does the UE transmit a MeasurementReport message within the next 10s?

    f_NR_Check_NoMeasurementReport(10.0, nr_Cell1, cr_NR_MeasurementReport_Any);

    //@siclog "Step  4" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in Table 8.1.3.2.1.3.2-1/2.

    f_NR_SetCellPowerList(v_CellPowerList_AtT0_T1);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 5" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event B1 with the measured RSRP value for NR Cell 1?

    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellId, tsc_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_EUTRA({cr_NR_MeasResultEUTRA_CellResults(v_EUTRA_CellId, cr_NR_MeasQuantityResultsEUTRA_Rsrp)}), cr_NR_MeasQuantityResults_RsrpRsrq);

  }//End of f_TC_8_1_3_2_1_TestBody()

After change

   function f_TC_8_1_3_2_1_TestBody() runs on NR_BASE_PTC

  {

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0_T1;

    var NR_AbsoluteCellPower_Type v_CellPower_T1_FR1 := -85;

    var PhysCellId v_PhysCellId;

    var ARFCN_ValueEUTRA v_ARFCN_ValueEUTRA;

    var EUTRA_PhysCellId v_EUTRA_CellId;

    var integer v_MeasurementBandwidth_int;

    var EUTRA_AllowedMeasBandwidth v_MeasurementBandwidth;

    var RSRP_Range v_RsrpThres;

    var Hysteresis v_Hysteresis := 4;

    var EventTriggerConfigInterRAT.reportAmount v_ReportAmount := r1;

    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList_NR;

    var template (value) MeasObjectParamsList_Type_EUTRA v_MeasuredObjectList_EUTRA;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    //Power levels of NR cell 1: Table 8.1.3.1.1.3.2-1, Table 8.1.3.1.1.3.2-2

    /*         |    FR1     |    FR2

     * -------------------------------

     *   T0    |    -85     |    FFS

     * -------------------------------

     *   T1    |    -85     |    FFS

     *

     */

    v_CoOrd_SysInfo := f_NR_WaitForCoOrd_SysInfoEUTRA (EUTRA);

    v_ARFCN_ValueEUTRA := v_CoOrd_SysInfo.Eutra[0].Arfcn;

    v_EUTRA_CellId := bit2int (v_CoOrd_SysInfo.Eutra[0].CellId);

    v_MeasurementBandwidth_int := v_CoOrd_SysInfo.Eutra[0].DL_Bandwidth;

    v_MeasurementBandwidth := f_NR5GC_IRAT_ConvertIntegerToEUTRA_AllowedMeasBandwidth(v_MeasurementBandwidth_int);

    v_CellPowerList_AtT0_T1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_T1_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    f_NR_SetCellPowerList(v_CellPowerList_AtT0_T1);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-RAT measurement and reporting for event B1

    //The UE transmits an RRCReconfigrationComplete message

    v_RsrpThres := 60;

    v_MeasuredObjectList_NR := {cs_MeasObjectId1(nr_Cell1)};

    v_MeasuredObjectList_EUTRA := {cs_MeasObjectId2_EUTRA(v_EUTRA_CellId, v_ARFCN_ValueEUTRA)};

    v_ReportConfigList := {cs_NR_ReportConfig_reportConfigInterRAT(tsc_NR_IdReportConfigId1, cs_NR_ReportConfigInterRAT_Trigger_eventB1_EUTRA(v_RsrpThres, v_Hysteresis, v_ReportAmount))};

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};

v_MeasConfig := f_NR_GenerateMeasurementConfigNR_EUTRA(v_ARFCN_ValueEUTRA, v_MeasurementBandwidth, -, v_MeasuredObjectList_NR, v_MeasuredObjectList_EUTRA, v_ReportConfigList, v_MeasIdConfigList, cs_MeasGapConfig_UE, cs_38508_QuantityConfig_EUTRA);  

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig);
    //@siclog "Steps  3" siclog@

    //Check: Does the UE transmit a MeasurementReport message within the next 10s?

    f_NR_Check_NoMeasurementReport(10.0, nr_Cell1, cr_NR_MeasurementReport_Any);

    //@siclog "Step  4" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in Table 8.1.3.2.1.3.2-1/2.

    f_NR_SetCellPowerList(v_CellPowerList_AtT0_T1);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 5" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event B1 with the measured RSRP value for NR Cell 1?

    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellId, tsc_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_EUTRA({cr_NR_MeasResultEUTRA_CellResults(v_EUTRA_CellId, cr_NR_MeasQuantityResultsEUTRA_Rsrp)}), cr_NR_MeasQuantityResults_RsrpRsrq);

  }//End of f_TC_8_1_3_2_1_TestBody()

4.8  Change 8
	Function name
	New template cs_MeasGapConfig_UE

	Reason for change
	Configured MeasGapConfig should be as 38.508-1 table 4.6.3-70 without conditiond GAP_FR1 or GAP_FR2 defined.

	Summary of change
	New template for MeasGapConfigUE only

	TTCN module
	

	MCC160 Comment
	


New template
    template (value) MeasGapConfig cs_MeasGapConfig_UE :=

  { /* @status     */

    gapFR2 := omit,

    gapFR1 := omit,

    gapUE := {

      setup := cs_38508_GapConfig

    }

  };

4.9 Change 9

	Function name
	fl_TC_8_1_3_2_1_NR5GC_EUTRA_TestBody

	Reason for change
	Max reference power of EUTRA cell 1 not sufficient for T1

	Summary of change
	1.Change max reference power of EUTRA cell to -73 (see change 10)

	TTCN module
	RRC_Measurement_NR5GC_EUTRA

	MCC160 Comment
	


Before change

  function f_TC_8_1_3_2_1_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B1 / Measurement of E-UTRA cells

    var integer v_EUTRA_CellPower_T0_FR1 := -93;

    f_EUTRA38_IRAT_Init(c31);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

    f_EUTRA38_SetCellPower(eutra_Cell1, v_EUTRA_CellPower_T0_FR1, v_EUTRA_CellPower_T0_FR1);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_3_2_1_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

    f_EUTRA_ReleaseAllCells();

  }

After change

   function f_TC_8_1_3_2_1_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B1 / Measurement of E-UTRA cells

    var integer v_EUTRA_CellPower_T0_FR1 := -93;

    f_EUTRA38_IRAT_Init(c31);
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -73);
    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

    f_EUTRA38_SetCellPower(eutra_Cell1, v_EUTRA_CellPower_T0_FR1, v_EUTRA_CellPower_T0_FR1);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_3_2_1_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

    f_EUTRA_ReleaseAllCells();

  }
4.10 Change 10

	Function name
	fl_TC_8_1_3_2_1_NR5GC_EUTRA_TestBody

	Reason for change
	Incorrect cell power level at time T1

	Summary of change
	correct cell power at T1 to -73 as per prose

	TTCN module
	RRC_Measurement_NR5GC_EUTRA

	MCC160 Comment
	


Before change

  function fl_TC_8_1_3_2_1_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var Frequency_IE_Type v_Frequency_IE_f1;

    var ChannelBandwidthDependency_Type v_ChBandDependency_f1;

    var integer v_Arfcn;

    var AllowedMeasBandwidth v_DL_AllowedMeasBandwidth;

    var integer v_DL_Bandwidth;

    var template (value) B28_Type v_EUTRA_CellId;

    var template (value) CoOrd_EUTRASysInfo_List_Type v_SysInfo_EUTRA;

    var integer v_EUTRA_CellPower_T1_FR1 := -93;

    v_EUTRA_CellId := int2bit(f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),28);

    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);

    v_ChBandDependency_f1 := f_EUTRA_BandDependentParam(v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth );

    v_DL_AllowedMeasBandwidth := v_ChBandDependency_f1.AllowedMeasBandwidth;

    v_DL_Bandwidth := f_EUTRA38_IRAT_ConvertEUTRA_AllowedMeasBandwidthToInteger(v_DL_AllowedMeasBandwidth);

    v_Arfcn := v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq_r9;

    v_SysInfo_EUTRA[0] := cs_CoOrdEUTRASysInfo(v_Arfcn, v_DL_Bandwidth, -, -, -,  v_EUTRA_CellId, -, -, -);

    f_EUTRA_NR_SendSysinfoCoOrd(NR, cs_EUTRA_NR_CoOrdSysInfo_Eutra(v_SysInfo_EUTRA));

    //@siclog "Step  4" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in Table 8.1.3.2.1.3.2-1/2.

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

    f_EUTRA38_SetCellPower(eutra_Cell1, v_EUTRA_CellPower_T1_FR1, v_EUTRA_CellPower_T1_FR1);

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);

  }

After change

   function fl_TC_8_1_3_2_1_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var Frequency_IE_Type v_Frequency_IE_f1;

    var ChannelBandwidthDependency_Type v_ChBandDependency_f1;

    var integer v_Arfcn;

    var AllowedMeasBandwidth v_DL_AllowedMeasBandwidth;

    var integer v_DL_Bandwidth;

    var template (value) B28_Type v_EUTRA_CellId;

    var template (value) CoOrd_EUTRASysInfo_List_Type v_SysInfo_EUTRA;

    var integer v_EUTRA_CellPower_T1_FR1 := -73;
    v_EUTRA_CellId := int2bit(f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),28);

    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);

    v_ChBandDependency_f1 := f_EUTRA_BandDependentParam(v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth );

    v_DL_AllowedMeasBandwidth := v_ChBandDependency_f1.AllowedMeasBandwidth;

    v_DL_Bandwidth := f_EUTRA38_IRAT_ConvertEUTRA_AllowedMeasBandwidthToInteger(v_DL_AllowedMeasBandwidth);

    v_Arfcn := v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq_r9;

    v_SysInfo_EUTRA[0] := cs_CoOrdEUTRASysInfo(v_Arfcn, v_DL_Bandwidth, -, -, -,  v_EUTRA_CellId, -, -, -);

    f_EUTRA_NR_SendSysinfoCoOrd(NR, cs_EUTRA_NR_CoOrdSysInfo_Eutra(v_SysInfo_EUTRA));

    //@siclog "Step  4" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in Table 8.1.3.2.1.3.2-1/2.

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

    f_EUTRA38_SetCellPower(eutra_Cell1, v_EUTRA_CellPower_T1_FR1, v_EUTRA_CellPower_T1_FR1);

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);
5. Execution Log Files

5.1 HiSilicon Balong 5000
The HiSilicon Balong 5000UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on NR5GC band n41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_1_3_2_1_LOG.html
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	[1]
	R5s191108:   Supporting information for addition of NR5GC inter-RAT test case 8.1.3.2.1 in FR1 path
                          This archive comprises:

                        -  html and text format execution log files for Keysight test platform



